

TO:

COLOUR GENIE

AUCKLAND DISTRICT USER GROUP

NEWSLETTER NO. 21

SEPTEMBER 1985

```

*****
**                                                                 **
**                                                                 **
**      DO YOU WANT TO BECOME PART OF A NEW ZEALAND FILM?      **
**                                                                 **
**      DO YOU LIVE IN AUCKLAND, AND ATTEND OUR MEETINGS?      **
**                                                                 **
**                                                                 **
**      For the sum of $40.00 per person, Mirage Films are looking **
**      for extras for their next film. The money will be paid as **
**      a donation, direct to the Users Group. Exactly when is   **
**      still to be advised. Please let me know at the meeting if **
**      you can help our Group earn a bit of money this way.     **
**                                                                 **
**      Remember, any money we can make to help the Group will   **
**      help us all in the long run.                               **
**                                                                 **
*****

```

At the last meeting (to which a good number of people turned up), the proposed Annual General Meeting was raised. This meeting, you may remember, is being held on Monday the 14th of October, and is mainly to elect new committee members and decide where the Group is going to go from here. Olwen Williams, our present President, will not be standing for re-election, and Ernie Roots, who was our Newsletter Editor, has since taken on more responsibility at work, and therefore has to work longer hours. He feels that he cannot devote the time he would like to the Group, so he will be standing down too.

Then there is Ken. The lad who has done everything for the Group, from running our meetings in an emergency (which turned out to be a regular thing over the past few months), to doing the joystick modifications, to helping me out with copying tapes and trying to find the colour problems in machine code tapes, to repairing the Colour Genies. Ken has turned his hand to anything. Unfortunately, he has been transferred to some dark and dismal place outside Christchurch (too far away even to attend the Christchurch Group meetings). This terrible event is happening on the 2nd of December, so Ken doesn't

even get to be at our Christmas party (which he arranged last year!) How we are going to replace him just doesn't bear thinking about.

So, there we are with three committee members that are standing down. We only actually had four of us, although we managed to snare both Andy and Chris, and hopefully they will both stand if nominated.

As time provides, Andy will carry on with repairing Colour Genies, and doing the joystick modifications, and replacing roms.

CARING FOR YOUR COLOUR GENIE

1. Don't clean the brown part of your Colour Genie with anything other than water - especially not meths.

2. Keith McGill suggests that if you are having trouble with the keys, that you undo the millions of screws holding the keyboard together, and clean the key contacts with something like Isopropal Alcohol. Some of his keys were sticking and this fixed it. Of course if you are like me and have less mechanical/technical ability than a wet fish, you are better off letting Andy do it.

NEW MEMBER

We have one new member to add to our Group this month -

Dominic Skinner, 6 Meadowcourt Drive, Papatoetoe

He and his entire family turned up at the meeting last month. We hope you enjoyed it, and we welcome you to the Group Dominic.

GREEN SCREEN MONITOR FOR SALE

The Elley family have a BMC Monitor for sale. It is in very good condition, but they find they get more out of their colour T.V., especially with the games, so the monitor never gets used. They are asking \$220.00 o.n.o. for it. Anyone interested, please Ph: 675 024.

SOFTWARE EXCHANGE

Doug Grenfell would like to swap Terry's Travels for any Invader-type

SOFTWARE/PROGRAMS SENT IN BY MEMBERS

The amount of you that have been sending in items for the newsletter lately has been terrific. Especially some of the newer members like Alistair Clark (whose program BOTTLES was very much enjoyed) and Ron Hart, whose programs are yet to appear. And too, we have the regulars who send so many programs in - Keith McGill (who is a veritable fountain of programs), Herman Philipson, who has done a heck of a lot of translation work that has taken many, many hours to do, Ted Doell, our Radio Ham man, who has sent in circuit diagrams and other interesting articles like the one on making a Monitor Speaker which will be coming up in one of the future newsletters. Then there is Don Edwards, who also obviously spends a lot of time at his Genie and sends us a lot of program printouts to prove it. Of course, there are others I haven't mentioned, but these four stand out.

To everyone who has contributed anything at all to the newsletter, a big thankyou. Without you, we couldn't keep the newsletter going.

Vola Huggins

Basic Basic for Parents & Beginners

Part 4

David Donaldson

Right then noses down & reading, here we go again. As promised we have a longer programme for you to get your teeth into this month & I think its about time we wrote the index for the series of utilities we have been writing. As the first programme is a bit more complex this month it will take a bit more thinking & explaining so here we go.

```
800 CLS
810 REM * Compound Interest
820 PRINT "To calculate the return for any amount of"
830 PRINT "principal for any interest rate & period"
840 PRINT "of time.
850 PRINT:PRINT:PRINT
860 INPUT "What is the principal ";P
870 INPUT "What is the rate of interest ";R
880 Input "What is the period of time in years ";T
890 PRINT "What is the frequency of interest crediting "
900 PRINT "1. Yearly
910 PRINT "2. Half yearly"
920 PRINT "3. Quarterly"
930 PRINT "4. Monthly"
940 INPUT F
950 IF F=1 THEN F=1
960 IF F=2 THEN F=2
970 IF F=3 THEN F=4
980 IF F=4 THEN F=12
990 X=0:Z=0
1000 Y=P*(R/100)/F
1100 F=P+Y
1110 X=X+1
1120 IF X<F GOTO 1000 ELSE 1130
1130 X=0:Z=Z+1
1140 IF Z<T GOTO 1000 ELSE 1150
1150 PRINT "Total compounded repayment will be ";P
1160 PRINT:PRINT:PRINT
1170 INPUT "Another calculation Y/N "Z$
1180 IF Z$= "Y" GOTO 800
1190 GOTO 10
```

Now lets have a look , well it certainly is a bit more complicated isn't it . OK now the first part is clear, CLS clear the screen please. A couple of PRINT statements telling us what the programme is for. A REM statement to 'tag' the programme then INPUT statements to get the necessary variable data into the beasties brainbox, each of the variables being represented by a letter representing its function ie 'P' for Principal, makes it a bit easier later on when working out what each one means when writing the alogrim (thats the part where all the calculations are done.) Then we get to the part where, in an IF statement we

find IF F=1 etc well you see that takes the key (1,2,3,4) which was pressed & converts it into the dividing factor for the computer to use. ie if we press key 4 for monthly interest crediting then the computer has to divide the interest by 12 not the key 4 we used. All the sort of stuff we have covered before eh but now the rest X=0:Z=0 right now this is for something new a LOOP yes I know a cowboy has one in the end of his lasso but thats not the one I'm talking about This is one the computer can do & whats more along with the IF/THEN statement is one of the two commands which give a computer its 'intelligence' .The IF/THEN command allows our computer to make a decision (wow) and the loop allows us to make it do a certain thing a specified number of times in this case the number of payments per year and also the number of years too. Here we have 'nested loops', sorry dropped you straight into the hard part didn't I. Well what happens is that we have TWO loops in this programme tagged 'X' & 'Z' .In the loop 'X' we make Gertie go back & do the calculation again for each time interest is credited per year THEN as if that's not enough we make her do it all over again for the number of years we invest our money for, 'Z' loop. Notice how first time around we add the interest to our original interest then each time after that it gets the next lot of interest added to the previous lot P=P+Y & all the time 'P' (principal) gets bigger. The reason for line 990 is to set both loops to zero so that if we want to do another calculation, previous data does not foul up our new calculations. Line 1130 sets our X loop to 0 for the next time around & the Z loop gets 1 added to keep track of the number of years we have processed. Then at the end a PRINT statement to show us the Golden Egg. In the last line we have a GOTO 10 this is for the overall index we will be writing next.

Right then to the index, just type in the following :-

```

10 CLS
20 REM Utility index
30 PRINT "Select which utility you require"
40 PRINT:PRINT:PRINT
41 PRINT "1. Convert Gal.to Litres"
42 PRINT "2. Convert Litres to Gal."
43 PRINT "3. Convert Centigrade to Farenheit"
44 PRINT "4. Convert Farenheit to Centigrade"
45 PRINT "5. Foreign Currency conversion"
46 PRINT "6. Average attendance"
47 PRINT "7. Compound interest"
90 INPUT X
95 ON X GOTO 110,210,310,410,610,710,810

```

Now I guess you can see commands we have used before so won't go into them again but however the line numbering is a bit strange isn't it ? Well I know I said to leave 10 between each line number however in a thing like an index where there will be no need to add to it later a bit of poetic licence can be taken & it saves line numbers too. Now we also have space to add extra lines as we add to the overall project, all we do is add a line 48 say

& the name of the programme & list its address to line 95.

Line 95 is a new command 'ON X GOTO' OK so we select '3' & press 'return' the computer knows to run along the list of addresses until the 3rd one then go to that location smart eh.

I thought that now we have an index we might as well tid up the whole listing a bit so just type in the following lines.

```
115 REM Gal to litres
215 REM Litres to Gal
315 REM Centigrade to Farenheit
415 REM Farenheit to Centigrade
190 GOTO 10
195 END
290 GOTO 10
295 END
390 GOTO 10
395 END
490 GOTO 10
495 END
695 GOTO 10
697 END
795 GOTO 10
797 END
```

This will tidy things up a bit if you have been typing the blocks in as they have been published. Now a bit of a word on saving your programmes to tape. If you have all this typed in so as not to run ANY risk of losing such a lot of typing a set way of saving will be essential. The idea is to have a 'backup' system of tapes. The tape you made last ie up to line 795 should be your most up to date copy so load it in, add the new material in this part then SAVE, but to a different tape & do it TWICE on the same tape. Now what you have is a complete listing twice on a tape & on the first tape the listing up to the last part. Next time you edit the listing LOAD from your most recent tape, edit, then SAVE to the PREVIOUS tape which now becomes the latest (remember twice) erasing the last copy as you do so. This way you play leapfrog with the two tapes The latest on one & as backup the last tape recorded. If you have a VERIFY command on your machine **Use It** both when loading & copying.

Thats the lot for this month, next month a couple of shorter programmes again so breathe easy.

```

10 '*****
20 '*
30 '*
31 '*      BIORHYTHM      CHART
32 '*      WITH SCREEN DUMP
33 '*
34 '*****
35 '
36 '
37 '
80 L$=LEFT$(L$,T)+CHR$(149)+RIGHT$(L$,T)
100 REM      BIORHYTHM
120 CLEAR 200:DEFINT K,L:DEFDBL B,J,M-Z
130 L=0:T=14:P=3.1415926535:CLS
140 PRINT TAB(15);"BIORHYTHM":PRINT
150 PRINT"ENTER BIRTH DATE"
160 GOSUB 500
170 GOSUB 600
180 JB=JD
190 PRINT:PRINT"ENTER START DATE FOR CHART"
200 GOSUB 500
210 GOSUB 600
220 JC=JD
230 IF JC>=JB THEN 270
240 PRINT"CHART DATE CAN'T BE EARLIER THAN"
250 PRINT"BIRTH DATE.  TRY AGAIN"
260 GOTO 150
270 FOR K=1 TO 1000:NEXT
280 GOSUB 700
300 N=JC-JB
310 V=23:GOSUB800:GOSUB 850
320 V=28:GOSUB800:GOSUB 850
330 V=33:GOSUB800:GOSUB 850
340 GOSUB 1000
350 PRINT C$;TAB(8);L$
360 JC=JC+1:L=L+1:IFL<18 THEN 300
370 PRINT:PRINT" <E>ND - SPACE TO CONTINUE - <P>RINT";
380 R$=INKEY$:IFR$="" THEN 380
390 IF R$="E" THEN 120 ELSE 395
395 IF R$="P" THEN 10000
400 L=0:GOTO 280
500 PRINT
505 INPUT"MONTH (1 TO 12)";M
510 M=INT(M):IFM<1 OR M>12 THEN 505
520 INPUT"DAY (1 TO 31)";D
530 D=INT(D):IFD<1 OR D>31 THEN 520
540 INPUT"YEAR ";Y
550 Y=INT(Y):IF Y<0 THEN 540
560 IF Y>99 THEN 580
570 Y=Y+1900:PRINT Y;"ASSUMED"
580 RETURN
600 W=FIX((M-14)/12)
610 JD=INT(1461*(Y+4800+W)/4)
620 B=FIX(367*(M-2-W*12)/12)
640 JD=JD+B
650 B=INT(INT(3*(Y+4900+W)/100)/4)
660 JD=JD+D-32075-B
670 RETURN
700 CLS
710 PRINT TAB(18);"BIORHYTHM"

```

ADAPTED FOR COLOUR GENIE
 BY ERNIE ROOFS - FROM A
 PUBLICATION BY T. RUGG &
 P. FELDMAN.

32 BASIC PROGRAMS FOR
 TRS/80 (LEVEL II)

```

720 PRINT
730 PRINTTAB(8);:PRINT"D O W N";TAB(22);"0";TAB(30);"U P"
740 PRINT TAB(8);
750 FOR K=1 TO T+T+1:PRINT CHR$(45);:NEXT:PRINT:RETURN
800 W=INT(N/V):R=N-(W*V)
810 RETURN
850 IFV<>23 THEN 900
860 L$=CHR$(32):FOR K=1 TO 5:L$=L$+L$:NEXT
870 L$=L$+LEFT$(L$,19)
880 L$=LEFT$(L$,T)+CHR$(124)+RIGHT$(L$,T)
890 IF V=23 THEN C$="P"
900 IF V=28 THEN C$="E"
910 IF V=33 THEN C$="I"
920 W=R/V:W=W*2*P
930 W=T*SIN(W):W=W+T+1.5
940 W=INT(W):A$=MID$(L$,W,1)
950 IF A$="P" OR A$="E" OR A$="*" THEN C$="*"
955 IF WQ1 THEN 980
957 IF W=T+T+1 THEN 990
960 L$=LEFT$(L$,W-1)+C$+RIGHT$(L$,T+T+1-W)
970 RETURN
980 L$=C$+RIGHT$(L$,T+T):RETURN
990 L$=LEFT$(L$,T+T)+C$:RETURN
1000 W=JC+68569:R=INT(4*W/146097)
1010 W=W-INT((146097*R+3)/4)
1020 Y=INT(4000*(W+1)/1461001)
1030 W=W-INT(1461*Y/4)+31
1040 M=INT(80*W/2447)
1050 D=W-INT(2447*M/80)
1060 W=INT(M/11):M=M+2-12*W
1070 Y=100*(R-49)+Y+W
1080 A$=STR$(D):W=LEN(A$)-1
1090 C$=MID$(A$,2,W)+"/"
1100 A$=STR$(M):W=LEN(A$)-1
1110 C$=C$+MID$(A$,2,W)+"/"
1120 A$=STR$(Y):W=LEN(A$)-1
1130 C$=C$+MID$(A$,W,2)
1140 RETURN
1200 '*****
1201 '*'
1202 '*'
1203 '* SCREEN DUMP ROUTINE
1204 '*'
1205 '*'
1206 '*****
1207 '
1208 '
10000 CLEAR:RESTORE
10005 P=17408
10010 FOR Z=0TO23
10020 FOR S=PTOP+39
10030 G=PEEK(S)
10040 LPRINT CHR$(G);:LPRINT" ";
10050 NEXT S
10060 LPRINT CHR$(128)
10070 P=P+40
10080 NEXT Z

```


WEST EDWARD ROOTS

2 : 12 : 1937

1 : 9 : 1985

BIORHYTHM

DOWN 0 UP

Date	Down	0	Up
11/9/85	I		PE
12/9/85	I		PEE
13/9/85	I		PEE
14/9/85	IP		EE
15/9/85	P		EE
16/9/85	P		EE
17/9/85	P		EE
18/9/85	P		IEE
19/9/85	P		IEE
20/9/85	P		IEE
21/9/85	P		IEE
22/9/85	P		IEE
23/9/85	PE		IEE
24/9/85	E		IEE
25/9/85	E		IEE
26/9/85	E		IEE
27/9/85	E		IEE
28/9/85	E		IEE
29/9/85	E		IEE
30/9/85	E		IEE
1/10/85	IP		IEE
2/10/85	PI		IEE
3/10/85	P		IEE
4/10/85	P		IEE
5/10/85	P		IEE
6/10/85	P		IEE
7/10/85	P		IEE
8/10/85	P		IEE
9/10/85	P		IEE
10/10/85	P		IEE
11/10/85	P		IEE
12/10/85	P		IEE
13/10/85	P		IEE
14/10/85	P		IEE
15/10/85	P		IEE
16/10/85	P		IEE
17/10/85	P		IEE
18/10/85	P		IEE
19/10/85	P		IEE

HAVE A NICE DAY

BIORHYTHM

PURPOSE

Did you ever have one of those days when nothing seemed to go right? All of us seem to have days when we are clumsy, feel depressed, or just cannot seem to force ourselves to concentrate as well as usual. Sometimes we know why this occurs. It may result from the onset of a cold or because of an argument with a relative. Sometimes, however, we find no such reason. Why can't we perform up to par on some of those days when nothing is known to be wrong?

Biorhythm theory says that all of us have cycles, beginning with the moment of birth, that influence our physical, emotional, and intellectual states. We will not go into a lot of detail about how biorhythm theory was developed (your local library probably has some books about this if you want to find out more), but we will summarize how it supposedly affects you.

The physical cycle is twenty-three days long. For the first 11½ days, you are in the positive half of the cycle. This means you should have a feeling of physical well-being, strength, and endurance. During the second 11½ days, you are in the negative half of the cycle. This results in less endurance and a tendency toward a general feeling of fatigue.

The emotional cycle lasts for twenty-eight days. During the positive half (the first fourteen days), you should feel more cheerful, optimistic, and cooperative. During the negative half, you will tend to be more moody, pessimistic, and irritable.

The third cycle is the intellectual cycle, which lasts for thirty-three days. The first half is a period in which you should have greater success in learning new material and pursuing creative, intellectual activities. During the second half, you are supposedly better off reviewing old material rather than attempting to learn difficult new concepts.

The ups and downs of these cycles are relative to each individual. For example, if you are a very self-controlled, unemotional person to begin with, your emotional highs and lows may not be very noticeable. Similarly, your physical and intellectual fluctuations depend upon your physical condition and intellectual capacity.

The day that any of these three cycles changes from the plus side to the minus side (or vice versa) is called a "critical day." Biorhythm theory says that you are more accident-prone on critical days in your physical or emotional cycles. Critical days in the intellectual cycle aren't considered as dangerous, but if they coincide with a critical day in one of the other cycles, the potential problem can increase. As you might expect, a triple critical day is one on which you are recommended to be especially careful.

Please note that there is quite a bit of controversy about biorhythms. Most scientists feel that there is not nearly enough evidence to conclude that biorhythms can tell you anything meaningful. Others believe that biorhythm cycles exist, but that they are not as simple and inflexible as the 23, 28, and 33 day cycles mentioned here.

Whether biorhythms are good, bad, true, false, or anything else is not our concern here. We are just presenting the idea to you as an interesting theory that you can investigate with the help of your TRS-80 computer.

HOW TO USE IT

The program first asks for the birth date of the person whose biorhythm cycles are to be charted. You provide the month and day as you might expect. For the year, you only need to enter the last two digits if it is between 1900 and 1999. Otherwise, enter all four digits.

Next the program asks you for the start date for the biorhythm chart. Enter it in the same way. Of course, this date cannot be earlier than the birth date.

After a delay of about a second, the program clears the screen and begins plotting the biorhythm chart, one day at a time. The left side of the screen displays the date, while the right side displays the chart. The left half of the chart is the "down" (negative) side of each cycle. The right half is the "up" (positive) side. The center line shows the critical days when you are at a zero point (neither positive or negative).

Each of the three curves is plotted with an identifying letter—P for physical, E for emotional, and I for intellectual.

```

400 CLS:PRINT "PRE-SCHOOL TYPING TRAINER"
410 A$="A":PRINT A$
420 B$="B":PRINT B$
430 C$="C":PRINT C$
440 D$="D":PRINT D$
450 E$="E":PRINT E$
460 F$="F":PRINT F$
470 G$="G":PRINT G$
480 H$="H":PRINT H$
490 I$="I":PRINT I$
500 J$="J":PRINT J$
510 K$="K":PRINT K$
520 L$="L":PRINT L$
530 M$="M":PRINT M$
540 N$="N":PRINT N$
550 O$="O":PRINT O$
560 P$="P":PRINT P$
570 Q$="Q":PRINT Q$
580 R$="R":PRINT R$
590 S$="S":PRINT S$
600 T$="T":PRINT T$
610 U$="U":PRINT U$
620 V$="V":PRINT V$
630 W$="W":PRINT W$
640 X$="X":PRINT X$
650 Y$="Y":PRINT Y$
660 Z$="Z":PRINT Z$
670 Y=0:X=0
680 K1$=INKEY$:IF K1$="" THEN GOTO 410
690 A=PEEK(&HFB40):IF A=1 THEN CLS:Y=0:X=0:GOTO 410
700 IF A=32 AND X=0 THEN PRINT@Y+X-1," "
710 X=X+5:GOTO 410
720 IF X+Y>35 THEN Y=Y+20
730 IF X+Y>75 THEN Y=Y+40
740 IF X+Y>115 THEN Y=Y+60
750 IF X+Y>155 THEN CLS:GOTO 400
760 IF K1$="A" THEN PRINT@Y+X,A$
770 IF K1$="B" THEN PRINT@Y+X,B$
780 IF K1$="C" THEN PRINT@Y+X,C$
790 IF K1$="D" THEN PRINT@Y+X,D$
800 IF K1$="E" THEN PRINT@Y+X,E$
810 IF K1$="F" THEN PRINT@Y+X,F$
820 IF K1$="G" THEN PRINT@Y+X,G$
830 IF K1$="H" THEN PRINT@Y+X,H$
840 IF K1$="I" THEN PRINT@Y-X,I$
850 IF K1$="J" THEN PRINT@Y+X,J$
860 IF K1$="K" THEN PRINT@Y+X,K$
870 IF K1$="L" THEN PRINT@Y+X,L$
880 IF K1$="M" THEN PRINT@Y-X,M$
890 IF K1$="N" THEN PRINT@Y+X,N$
900 IF K1$="O" THEN PRINT@Y+X,O$
910 IF K1$="P" THEN PRINT@Y+X,P$
920 IF K1$="Q" THEN PRINT@Y+X,Q$
930 IF K1$="R" THEN PRINT@Y+X,R$
940 IF K1$="S" THEN PRINT@Y-X,S$
950 IF K1$="T" THEN PRINT@Y-X,T$
960 IF K1$="U" THEN PRINT@Y+X,U$
970 IF K1$="V" THEN PRINT@Y+X,V$
980 IF K1$="W" THEN PRINT@Y+X,W$
990 IF K1$="X" THEN PRINT@Y+X,X$
1000 IF K1$="Y" THEN PRINT@Y+X,Y$
1010 IF K1$="Z" THEN PRINT@Y+X,Z$
1020 X=X+1
1030 GOTO 410

```

PRE-SCHOOL TYPING TRAINER

- prints the alphabet in large letters.
 - five rows of 8 letters.
 - Use <CLEAR> to start again.
 - Use <←> to backspace within a line
- All "-" are four characters including spaces.
 Designed by Keith McGill, used successfully by his 2-75 year old grandson.

VISITAPE

Do you want to see what's on your tapes? Try this wee program:

```
10 FOR I=0 TO 12:READ N:POKE&H7800+I,N:NEXT:CLS:CALL 7800
20 DATA 205,76,2,33,0,68,205,237,1,119,44,24,249
```

The Assembly code for this routine is:

```
7800    CD 4C 02    CALL 024C        ;FIND THE START
7803    21 00 44    LD HL,4400H      ;SET TOP OF SCREEN MEM
7806    CD ED 01    CALL 01ED        ;GET A BYTE FROM TAPE
7809    77          LD (HL),A        ;DISPLAY THE BYTE
780A    2C          INC L            ;USE TOP 6 LINES
780B    18 F9      JR 7806          ;GET NEXT BYTE
```

RUN the program and start the tape of the program you want to see. The content of your tape is displayed across the top six lines of the screen but it doesn't load into memory. When the tape read has finished, do a ReSeT and the program is ready to RUN again. This program is useful for checking the tapes of which you have lost the index and for finding the best volume setting for data tapes. Start with a slightly high volume and then back off the volume slowly. You will see the load suddenly disintegrate into garbage. The best setting seems to be just above this point.

Do you like it? You can add this into the WP2 word processor quite easily and without any penalty on memory requirements. Data tapes load as usual. Make the following alterations to lines 384 and 386. The areas of change are underlined for your convenience.

```
384 FOR A=0TO62: ...
```

```
386 DATA 27,203,122,192,0,62,0,254,0,40,7,70,35,126,35,229,102
,111,213,80,58,88,65,254,0,32,10,205,237,1,6,68,2,12,119,
24,10,1,255,0,205,96,0,126,205,31,2,66,209,35,58,70,64,254
,0,40,199,16,215,225,35,24,193
```

If any out-of-town members want to contact me directly regarding the WP2 word processor or these amendments, my address is:

47 Marsden St
Takapuna
Auckland 9

- Allan Clarke

A HISTORICAL(?) PROGRAM

The listing for the cost analysis program is from the EACA demonstration tape that came with the original System 80 family. The listing is as was, including the Hong Kong english!

- Allan Clarke

WORD PROCESSOR NOTES

This month I shall discuss another way of inserting printer control codes into your text and describe how you can add your own routine into the program.

The insertion of printer control codes is done under the Insert function while in the Edit mode. Having typed all of your text and prettied it up with Adjust and Justify, Edit the line you want the control code in. Press <I> to go into Insert mode and move the cursor to the right position with the right and left arrow keys. There are two sets of control codes, those starting with ESC or CHR\$(27) and those starting with CTRL. ESC codes are entered by pressing the up arrow and then the required letter or number. For example, the italic mode can be selected on the Super 5 by ESC 4 (press up arrow and then 4). Now while staying in the Edit/Insert mode, move the cursor to the place where you want the italics to stop. Now press the up arrow and then 5, (ESC 5). Got it? Press <RETURN> to end the insertion.

CTRL codes are inserted in the same way but instead of using the up arrow, you HOLD DOWN the <SHIFT> and down arrow keys together and press the required letter. For example, to go into wide print mode, hold down the <SHIFT> and down arrow keys and press the <N> key (CTRL N). Most printers use this code and usually drop back into the normal print at the end of the line. Your printer manual will contain more details of the ESC and CTRL codes.

It does not pay to watch what happens on the screen when you are inserting these codes as the display seems to go crazy! However, the result works on the printer and I have been able to save the edited version on tape.

There is a place reserved in the Wordmate WP2 for your own routine such as that described by Keith McGill in the June Newsletter. The COMMAND used is Y (for "Yours", see!) and the routine is located at line 398. This contains a GOTO 24 in the standard version. You would usually end your routine with a GOTO 20. Be warned that the memory left after CLEARing 20000 for string space is very small indeed and you will need to reduce the amount CLEARed by the number of bytes in the routine. Also be warned not to edit any lines before line 24. This may be fatal for the program!

- Allan Clarke

NEW "NEW ROMS"

I have been working on some amendments to ROMs 1 and 4 with Ken Hynds invaluable assistance. These use up some of the spare memory in the Genie. The changes include: 1. A built-in driver for serial printers, called up by two POKEs. 2. An improved key debounce routine which also speeds up key response. 3. Changing the Shift-lock key into a Caps-lock and either booting-up in caps or lower-case mode. 4. Starting the Genie up in white. 5. A possible permanent fix for the PRINT#-1/INPUT#-1 bug. These changes are in the final testing stages and should be available later on next month at a price yet to be determined. The cost will be for the EPROM blowing only.

- Allan Clarke

```

10 CLS
20 PRINTTAB(11);"WORD SEARCH PUZZLE"
30 REM:THIS VERSION IS DESIGNED FOR A PRINTER.
40 REM:ADAPTED FROM DAVID H AHL 'MORE BASIC COMPUTER GAMES'
41 REM:BY A.K.McGILL
50 PRINT:PRINT:PRINT
60 PRINT
70 PRINT:PRINT
80 CLEAR 3000
90 TW=40
100 INPUT"DO YOU WANT A PRINTOUT";X$
110 INPUT"WIDTH";W:MD=W
120 IF W*2<=TW THEN 140
130 PRINT"THAT WILL NOT FIT IN ";TW;" COLUMNS":GOTO 110
140 IF W<1 THEN 110
150 INPUT"THE LENGTH ";L:IF L>W THEN MD=L
160 IF L<1 THEN 150
170 INPUT"WHAT IS THE MAXIMUM NUMBER OF WORDS IN THE PUZZLE";M
180 IF M>=2 THEN 200
190 PRINT"SORRY! - THERE MUST BE AT LEAST 2 WORDS":GOTO 170
200 PRINT
210 DIM A$(L,W),W$(M)
220 DIM W(M,3),DXY(8,2),DD(30)
230 PRINT"ENTER A HEADING FOR THE PUZZLE"
240 PRINT(";"TW;" CHARACTERS MAXIMUM)"
250 INPUT XY$
260 PRINT"OK - ENTER A WORD AT EACH QUESTION MARK"
270 PRINT"TO REDO THE PREVIOUS WORD, TYPE A HYPHEN"
280 PRINT"WHEN YOU RUN OUT OF WORDS,TYPE A STOP"
290 FOR I=1 TO M
300 INPUT T$:IF T$="-" THEN I=I-1:PRINT"REDO ";W$(I);"...":GOTO 300
310 IF T$="." THEN M=I-1:GOTO 500
320 IF LEN(T$)=0 THEN PRINT"INPUT ERROR;REDO!":GOTO 300
330 J=1
340 TE$=MID$(T$,J,1):IF TE$>="a" AND TE$<="z" THEN 410
350 IF TE$<"A" OR TE$>"Z" THEN 370
360 T$=LEFT$(T$,J-1)+CHR$(ASC(MID$(T$,J,1))+32)+RIGHT$(T$,LEN(T$)-J):GOTO 410
370 IF TE$=T$ THEN T$="":GOTO 320
380 IF J=LEN(T$) THEN T$=LEFT$(T$,J-1):GOTO 420
390 IF J=1 THEN T$=RIGHT$(T$,LEN(T$)-1):J=J-1:GOTO 410
400 T$=LEFT$(T$,J-1)+RIGHT$(T$,LEN(T$)-J):J=J-1
410 J=J+1:IF J<=LEN(T$) THEN 340
420 PRINT"-";T$;"-"
430 IF LEN(T$)<=MD THEN 450
440 PRINT"THAT'S TOO LONG,I'M AFRAID!";" TRY ANOTHER ONE!":GOTO300
450 FOR IZ=1 TO I-1:IF W$(IZ)<>T$ THEN NEXT:GOTO 470
460 PRINT"YOU'VE ENTERED THAT ALREADY!TRY ANOTHER":GOTO300
470 W$(I)=T$
480 NEXT I
490 PRINT"THAT'S IT! ";M;" WORDS!"

500 PRINT"NOW LET ME PONDER THIS...."
510 FOR I=1 TO M-1
520 FOR J=I+1 TO M
530 IF LEN(W$(I))<LEN(W$(J)) THEN HZ$=W$(I):W$(I)=W$(J):W$(J)=HZ$
540 NEXT:NEXT
550 FOR I=1 TO 8:READ DXY(I,1),DXY(I,2):NEXT
560 FOR I=1 TO 28:READ DD(I):NEXT
570 DATA 0,1,1,1,1,0,1,-1,0,-1,-1,-1,-1,0,-1,1
580 DATA 2,4,6,8,2,4,6,8,2,4,6,8,2,4,6,8,2,4,6,8,2,4,6,8,2,4,6,8,1,3,5,7
590 FOR I=1 TO M
600 LN=LEN(W$(I))
610 NT=0
620 SD=DD(RND(28))

```

```

630 SX=RND(W):X1=SX+(LN-1)*DXY(SD,1):IFX1<10R X1>W THEN 620
640 SY=RND(L):X1=SY+(LN-1)*DXY(SD,2):IF X1<10R X1>L THEN 620
650 NT=NT+1:IF NT<>W*L*2 THEN 700
660 PRINT"COULDN'T FIT '";W$(I);" IN THE PUZZLE "
670 PRINT"DO YOU WANT ME TO START AGAIN?";:INPUT A$
680 IF LEFT$(A$,1)="y" THEN 590
690 W$(I)="" :GOTO 780
700 J=SY:K=SX
710 FOR P=1 TO LN
720 IF LEN(A$(J,K)) AND A$(J,K)<>MID$(W$(I),P,1) THEN 620
730 J=J+DXY(SD,2):K=K+DXY(SD,1):NEXT P
740 J=SY:K=SX
750 FOR P=1 TO LN:A$(J,K)=MID$(W$(I),P,1)
760 J=J+DXY(SD,2):K=K+DXY(SD,1):NEXT
770 W(I,1)=SX:W(I,2)=SY:W(I,3)=SD
780 NEXT I
790 FOR I=1 TO L
800 FOR J=1 TO W
810 IF A$(I,J)="" THENA$(I,J)=CHR$(RND(26)+96)
820 NEXT: NEXT
830 FOR I=1 TO M-1:FOR J=I+1 TO M
840 IF W$(I)<=W$(J) THEN 870
850 HZ$=W$(I):W$(I)=W$(J):W$(J)=HZ$
860 FOR K=1 TO 3: HZ=W(I,K):W(I,K)=W(J,K):W(J,K)=HZ:NEXT K
870 NEXT J:NEXT I
880 INPUT"HOW MANY COPIES DO YOU WANT";N
890 PRINT"FOR EACH COPY,HIT RETURN TO BEGIN PRINT"
900 FOR C=1 TO N:GOSUB 910:NEXT:GOTO 1090
910 INPUT A$:LPRINT
920 T=(TW-2*W)/2:LPRINT
930 LPRINT
940 LPRINTTAB((TW-LEN(XY$))/2);XY$
950 LPRINT:LPRINT
960 FOR J=1 TO L:LPRINTTAB(T);
970 FOR K=1 TO W:IF A$(J,K)=". " THEN LPRINT". ";:GOTO 990

980 LPRINTCHR$(ASC(A$(J,K))-32);" ";
990 IF K=W THEN LPRINT
1000 NEXT:NEXT
1010 LPRINT:LPRINT
1020 LPRINT"FIND ";M;"HIDDEN WORDS IN THE ABOVE PUZZLE"
1030 LPRINT
1040 FOR J=1 TO M:IF LEN(W$(J))=0 THEN 1070
1050 IF POS(0)+LEN(W$(J))>TW-2 THEN LPRINT
1060 LPRINTW$(J),
1070 NEXT:PRINT
1080 RETURN
1090 IF LEFT$(X$,1)="Y" OR LEFT$(X$,1)="y" THEN 1110
1100 END
1110 REM
1120 FOR I=1 TO L:FOR J=1 TO W:A$(I,J)=". ":NEXT J:NEXT I
1130 FOR I=1 TO M
1140 LN=LEN(W$(I)):J=W(I,2):K=W(I,1)
1150 FOR P=1 TO LN
1160 A$(J,K)=MID$(W$(I),P,1)
1170 J=J+DXY(W(I,3),2):K=K+DXY(W(I,3),1):NEXT P
1180 NEXT I
1190 XY$="HERE IS THE ANSWER KEY:"
1200 GOSUB 920
1210 CLS:PRINT$520,"TO INPUT ANOTHER PUZZLE,PRESS 'Q' "
1220 K$=INKEY$:IF K$="" THEN 1220 ELSE IF K$="Q" THEN RUN ELSE END
1230 END
1240 Z=INT(RND(Z)):RETURN

```